LHeC revolution frequencies

John Jowett

In[286]:= **DateString[]**

Out[286]= Wed 19 Sep 2007 13:30:32

Setup

Values of parameters

Rule for some constants in the problem (also takes care of a temporary problem with package, needs adapting for *Mathematica* V6).

In[238]:=
```
constsN = {m_p → ProtonMass, m_e → ElectronMass, ppmax → 7 TeV / c,
  c → SpeedOfLight, Cp -> 26 658.8832 Meter, alphacLHC → 0.0003216612973, λ → 1}
```

Out[238]= $\{m_p \to 1.67262\times 10^{-27}\ \text{Kilogram},\ m_e \to 9.10938\times 10^{-31}\ \text{Kilogram},\ \text{ppmax} \to \frac{7\ \text{ElectronVolt Tera}}{c},\ c \to \frac{299\,792\,458\ \text{Meter}}{\text{Second}},\ \text{Cp} \to 26\,658.9\ \text{Meter},\ \text{alphacLHC} \to 0.000321661,\ \lambda \to 1\}$

Circumference relations for protons and electrons

Velocity of a general ion, momentum p, mass m, charge Q

In[218]:= $\text{ionVelocity}[p\_, \{m\_, Q\_\}] := c \sqrt{1 - \frac{c^4 m^2}{c^4 m^2 + c^2 p^2 Q^2}}$

Revolution period of protons in LHC ring, of circumference C_p. Introduce small parameter λ (of order $1/\gamma$ for protons)

In[219]:= $\text{Tp}[\text{Cp}\_, \text{pp}\_] = \text{Series}\left[\frac{\text{Cp}}{\text{ionVelocity}[\text{pp}, \{\lambda\, m_p, 1\}]}, \{\lambda, 0, 3\}\right]$

Out[219]= $\frac{\text{Cp}}{c} + \frac{c\ \text{Cp}\ m_p^2 \lambda^2}{2\ \text{pp}^2} + O[\lambda]^4$

Revolution period of electrons in LHeC ring, of circumference C_e. Extra small parameter λ (of order m_e / m_p)

In[220]:= $\text{Te}[\text{Ce}\_, \text{pe}\_] = \text{Series}\left[\frac{\text{Ce}}{\text{ionVelocity}[\text{pe}, \{\lambda^2 m_e, 1\}]}, \{\lambda, 0, 3\}\right]$

Out[220]= $\frac{\text{Ce}}{c} + O[\lambda]^4$

Suppose that the circumferences have been adjusted to match at top momentum ppmax, pemax

In[221]:= `Tp[Cp, ppmax] == Te[Ce, pemax]`

Out[221]= $\frac{Cp}{c} + \frac{c\,Cp\,m_p^2\,\lambda^2}{2\,ppmax^2} + O[\lambda]^4 == \frac{Ce}{c} + O[\lambda]^4$

This fixes the circumference of the electron ring.   To this order of calculation it doesn't matter what pemax is.

In[222]:= `Cerule = Flatten[Simplify[Solve[Normal[Tp[Cp, ppmax] == Te[Ce, pemax]], Ce]]]`

Out[222]= $\left\{Ce \to Cp + \frac{c^2\,Cp\,m_p^2\,\lambda^2}{2\,ppmax^2}\right\}$

At other momenta, the ratio of proton to electron revolution period is independent of the electron momentum in this approximation

In[223]:= $\frac{\text{Tp[Cp, pp]}}{\text{Te[Ce, pe]}}$ `/. Cerule`

Out[223]= $1 + \left(\frac{c^2\,m_p^2}{2\,pp^2} - \frac{c^2\,m_p^2}{2\,ppmax^2}\right)\lambda^2 + O[\lambda]^4$

In[224]:= `peRevolutionPeriodRatio =` $\frac{\text{Tp[Cp, pp]}}{\text{Te[Ce, pe]}}$ `- 1 /. Cerule`

Out[224]= $\left(\frac{c^2\,m_p^2}{2\,pp^2} - \frac{c^2\,m_p^2}{2\,ppmax^2}\right)\lambda^2 + O[\lambda]^4$

In[225]:= `peRevolutionPeriodRatioN =`
`Normal[`$\frac{\text{Tp[Cp, pp]}}{\text{Te[Ce, pe]}}$ `- 1] //. Flatten@{Cerule, constsN, pp → ppTeV TeV / c} // N //`
`ToFundamentalSI`

Out[225]= $-8.9832\times10^{-9} + \frac{4.40177\times10^{-7}}{ppTeV^2}$

In[226]:=
```
peRevolutionPeriodRatioN =
 Normal[peRevolutionPeriodRatio] //. Flatten@{constsN, pp → ppTeV TeV / c} // N //
  ToFundamentalSI
```

Out[226]= $-8.9832\times10^{-9} + \frac{4.40177\times10^{-7}}{ppTeV^2}$

In[227]:= `Plot[peRevolutionPeriodRatioN, {ppTeV, 0.45, 7}, AxesLabel → {"`p_p`/TeV/c", "`T_p/T_e`-1"}]`

Out[227]=


The difference in revolution periods is

In[228]:=
```
peRevolutionPeriodDifference = Tp[Cp, pp] - Te[Ce, pe] /. Cerule
```

Out[228]= $\left(\frac{c\,\text{Cp}\,m_p^2}{2\,\text{pp}^2} - \frac{c\,\text{Cp}\,m_p^2}{2\,\text{ppmax}^2}\right)\lambda^2 + O[\lambda]^4$

Numerically, multiplying by c to get a length

In[234]:=
```
peRevolutionPeriodDifferenceMeterN =
  Normal[(c peRevolutionPeriodDifference)] / Meter /. pp → ppTeV TeV / c //. constsN //
   ToFundamentalSI // Simplify
```

Out[234]= $-0.000239482 + \frac{0.0117346}{\text{ppTeV}^2}$

At LHC injection this reaches a maximum of

In[235]:=
```
peRevolutionPeriodDifferenceMeterN /. ppTeV → 0.45
```

Out[235]= 0.0577093

In[237]:=
```
LogPlot[peRevolutionPeriodDifferenceMeterN,
 {ppTeV, 0.45, 7}, FrameLabel → {"p_p/(TeV/c)", "c(T_p-T_e)/m"},
 Axes → False, Frame → True, PlotRange → {{0, All}, {0, All}}]
```

Out[237]=



This is *twice* the distance by which the encounter points of the two beams would move on each turn if the RF frequencies were adjusted to keep each beam on its central orbit.

Equalizing revolution period with RF frequency

Revolution periods could be equalised by equalising the two beams' RF frequencies. If both are adjusted to keep the protons on their central orbit (which I would think is preferable to moving both beams), the electron momentum will increase by a fraction δ to lengthen its orbit. Assume for now that both rings have the same momentum compaction

In[253]:=
```
Tp[Cp, pp] == Te[Ce (1 + alphacLHC δ), pe] /. Cerule
```

Out[253]= $\frac{\text{Cp}}{c} + \frac{c\,\text{Cp}\,m_p^2\,\lambda^2}{2\,\text{pp}^2} + O[\lambda]^4 == \frac{\text{Cp}\,(1 + \text{alphacLHC}\,\delta)}{c} + \frac{c\,\text{Cp}\,m_p^2\,(1 + \text{alphacLHC}\,\delta)\,\lambda^2}{2\,\text{ppmax}^2} + O[\lambda]^4$

```
δe = Series[
   δ /. First@Solve[Normal[Tp[Cp, pp] == Te[Ce (1 + alphacLHC δ), pe] /. Cerule /. Cerule], δ],
   {λ, 0, 2}] // Simplify
```

Out[255]= $\frac{c^2\, m_p^2 \left(-pp^2 + ppmax^2\right) \lambda^2}{2\, alphacLHC\, pp^2\, ppmax^2} + O[\lambda]^3$

In[264]:= `δeN = Normal[peδ] /. pp → ppTeV TeV/c //. constsN // ToFundamentalSI // Simplify`

Out[264]= $-0.0000279275 + \frac{0.00136845}{ppTeV^2}$

In[265]:= `LogPlot[δeN, {ppTeV, 0.45, 7}, FrameLabel → {"pp/(TeV/c)", "δe"}, Axes → False, Frame → True, PlotRange → {{0, All}, {0, All}}]`

Out[265]=



This reaches a maximum value at injection

In[266]:= `δeN /. ppTeV → 0.45`

Out[266]= 0.00672984

which, with a supposed maximum dispersion of about 2 m would give a maximum orbit displacement of 13 mm.   LEP was able to handle this. However a more restrictive limit arises from the change in damping partition numbers. Guessing a value $J_\epsilon'(\delta) = -J_x'(\delta) = 400$, and allowing a maximum change of 0.5 (this limits the horizontal emittance increase to a factor 2), the lower limit to this would be at

```
Solve[400 δeN == 0.5, ppTeV]
```

Out[268]= {{ppTeV → -1.03481}, {ppTeV → 1.03481}}

Thus, with this approach, it would not be possible to make collisions below about 1 TeV *proton* energy.

Horizontal damping would be lost completely at

In[273]:= `Solve[400 δeN == 1, ppTeV]`

Out[273]= {{ppTeV → -0.735753}, {ppTeV → 0.735753}}

The blow up factor for the horizontal emittance would be

In[276]:=
```
Plot[1 / (1 - 400. δeN), {ppTeV, 0.45, 7}, FrameLabel → {"p_p/(TeV/c)", "δ_e"},
  Axes → False, Frame → True, PlotRange → {{0, All}, {0, 3}}]
```

Out[276]=


Estimates of luminosity for proton energies below 7 TeV would have to take this into account. Remember that this is independent of the electron energy.

At this point one could have recourse to shifting the proton beam inwards, but since the momentum compaction of the proton ring will be similar (and the protons are already quite relativistic at injection), they will have to take up the remaining shift in δ. Usually we limit the orbit shift in the LHC to 1 or 2 mm in collision conditions so this will not be enough.

Equalizing circumference with electron closed orbit

To compensate the circumference difference without displacing the proton beam orbit in the LHC, one could try to lengthen the electron ring circumference with a "pretzel" like orbit, i.e., long closed bumps around the arcs of the ring. This could be done with orbit corrector magnets. If the horizontal tune (in the arc FODO cells of the electron ring) is $Q_x = 70$ (as a guess) and the peak pretzel amplitude is x_p then (LEP Chamonix workshop 1993, Eq 47.1) we have an approximate formula for the increase in orbit length

In[280]:= $\text{dCepretzel[Qx\_, xp\_]} = \frac{8\,\text{Qx}^2\,\text{xp}^2}{\text{Ce}}$ /. Cerule

Out[280]= $$\frac{8\,\text{Qx}^2\,\text{xp}^2}{\text{Cp} + \frac{c^2\,\text{Cp}\,m_p^2\,\lambda^2}{2\,\text{ppmax}^2}}$$

In[281]:=
```
dCepretzel[70, xp] //. constsN // ToFundamentalSI
```

Out[281]= $$\frac{1.47043\,\text{xp}^2}{\text{Meter}}$$

We might consider a pretzel amplitude up to 20 mm:

In[285]:=
```
Plot[dCepretzel[70, xpN Meter]/Meter //. constsN // ToFundamentalSI,
  {xpN, 0, 0.02}, AxesLabel → {"x_p/m", "(C(x_p)-C(0))/m"}]
```

Out[285]=



But we still cannot gain the 0.05 m or so that we are looking for

---

Conclusion

If the LHeC electron ring is built with an aperture comparable to LEP, then ep collisions are possible with proton energies reduced to roughly 1 TeV (with the parameter guesses above). Below this energy one will need to ensure sufficient separation at all beam-beam encounters to avoid dangerous modulational effects.